Worksheet 9

9.1 Try to walkthrough the following program and write down the expected results. Key-in the program and compare the results after execution. This program helps you to understand the use of two-dimensional arrays.

Step 1: *Create a form with two command buttons, three labels, two text boxes, two arrays of text boxes and two arrays of labels according to the properties table below*

Properties Table

Object	**Property**	**Setting**
Form	Name Caption	frmTable Time Table
Label	Name Caption	lblMessage
Command Button	Name Caption	cmdCompute &Average and deviation
Command Button	Name Caption	cmdClear &Clear Table
Command Button	Name Caption	cmdExit &Exit
Array of labels	Name Caption Index	lblPeriod (empty) 0
Array of labels	Name Caption Index	lblDay (empty) 0
Array of text boxes	Name Text Index Background color ForeColor	txtPlace (empty) 0 &H80000000& red

Layout



Only the first elements of the arrays are created during design time and the remaining elements are created at run time..

Step 2 : *Declare a form array Place, two constants NumberDays and NumberPeriods. Add codes for the procedure Form_Load*

```
Const NumberDays = 5
Const NumberPeriods = 7
Dim Place(1 To NumberDays, 1 To NumberPeriods) As Single

Private Sub Form_Load()
    Dim i As Integer, Index As Integer
    Dim Day As Integer, Period As Integer

    lblPeriod(0).Caption = "1"
    For i = 1 To NumberPeriods - 1
        Load lblPeriod(i)
        lblPeriod(i).Left = lblPeriod(0).Left + i * (lblPeriod(0).Width + 100)
        lblPeriod(i).Caption = Str(i + 1)
        lblPeriod(i).Visible = True
    Next i

    lblDay(0).Caption = "Mon"
    For i = 1 To NumberDays - 1
        Load lblDay(i)
        lblDay(i).Top = lblDay(0).Top + i * (lblDay(0).Height + 100)
        Select Case i
            Case 1: lblDay(i).Caption = "Tue"
            Case 2: lblDay(i).Caption = "Wed"
            Case 3: lblDay(i).Caption = "Thu"
            Case 4: lblDay(i).Caption = "Fri"
            Case 5: lblDay(i).Caption = "Sat"
            Case 6: lblDay(i).Caption = "Sun"
        End Select
        lblDay(i).Visible = True
    Next i

    cmdCompute.Top = lblDay(NumberDays - 1).Top + lblDay(0).Height + 200
    cmdClear.Top = cmdCompute.Top
    cmdExit.Top = lblDay(NumberDays - 1).Top + lblDay(0).Height + 200
    cmdExit.Left = lblPeriod(NumberPeriods - 1).Left
                                + lblPeriod(0).Width - cmdExit.Width

    frmTable.Height = cmdExit.Top + cmdExit.Height + 1000
    frmTable.Width = lblPeriod(NumberPeriods - 1).Left + lblPeriod(0).Width + 500
    lblMessage.Width = frmTable.Width - 500
    lblMessage.Caption = "This program reads the information of the weekday, " _
        & " period and room from a file and prints the time table."

    txtPlace(0).Width = lblPeriod(0).Width
    txtPlace(0).Visible = False

    For Day = 1 To NumberDays
        For Period = 1 To NumberPeriods
            Index = (Day - 1) * NumberPeriods + Period - 1
            If Index <> 0 Then
                Load txtPlace(Index)
            End If
            txtPlace(Index).Left = lblPeriod(Period - 1).Left
            txtPlace(Index).Top = lblDay(Day - 1).Top
        Next Period
    Next Day

End Sub
```

Step 3 : *Add codes for the events*
cmdCompute_Click()
cmdClear_Click()
cmdExit

Codes for **cmdExit**

```
Private Sub cmdExit_Click()
    End
End Sub
```

Codes for **cmdCompute**

```
Private Sub cmdCompute_Click()
    Dim Day As Integer, Period As Integer, Room As Integer
    Dim Index As Integer

    Open App.Path & "\InFile9-1.dat" For Input As #1

    For Day = 1 To NumberDays
        For Period = 1 To NumberPeriods
            Place(Day, Period) = 0
        Next Period
    Next Day

    Do While Not EOF(1)
        Input #1, Day
        If Not EOF(1) Then
            Input #1, Period, Room
            Place(Day, Period) = Room
        End If
    Loop

    Call PrintTable(Place())
    Close #1

End Sub
```

Codes for **cmdClear**

```
Private Sub cmdClear_Click()
    Dim Index As Integer
    Dim Day As Integer, Period As Integer

    txtPlace(0).Visible = False
    For Day = 1 To NumberDays
        For Period = 1 To NumberPeriods
            Index = (Day - 1) * NumberPeriods + (Period - 1)
            If Place(Day, Period) <> 0 Then
                If Index <> 0 Then
                    txtPlace(Index).Visible = False
                End If
            End If
        Next Period
    Next Day
End Sub
```

Step 4 : *Add a procedure PrintTable*

```
Private Sub PrintTable(Place() As Single)
    Dim Index As Integer
    Dim Day As Integer, Period As Integer

    For Day = 1 To NumberDays
        For Period = 1 To NumberPeriods
            Index = (Day - 1) * NumberPeriods + (Period - 1)
            If Place(Day, Period) <> 0 Then
                txtPlace(Index).Text = Place(Day, Period)
                txtPlace(Index).Visible = True
            End If
        Next Period
    Next Day
End Sub
```

Step 5 : *Execution*

Click the **Print Time Table** command button.
Click the **Clear table** command button.
Try to modify the data file and execute again.
Terminate the program by clicking the **Exit** button.

Data file ”Infile9-1.dat”:



```
1   1   101
1   3   111
2   1   202
2   4   203
3   4   301
3   5   204
4   1   101
5   1   303
5   5   201
5   7   203
1   2   111
4   6   222
4   7   123
```